
@ECHO OFF 

REM =========================================================================== 

REM  HISTORIK.BAT               1.009                               12. 4.2025 

REM =========================================================================== 

REM  Startprogramm zu HISTORIK in QBASIC 

REM =========================================================================== 

REM  Besonderheiten: 

REM  Bei Sprungmarken beachtet MS-DOS 3.30 nur die ersten 8 Buchstaben! 

REM =========================================================================== 

REM  Bearbeitung: 29.11.1994 - 12. 4.2025 Norbert Südland 

REM =========================================================================== 

 

REM Ein temporäres Verzeichnis wird eingerichtet und QBASIC installiert: 

REM -------------------------------------------------------------------- 

REM Es besteht auch die Möglichkeit, HISTORIKTEMP direkt zu vorzugeben! 

REM ------------------------------------------------------------------- 

IF NOT x%DOSDRV%x==xx SET SYSTEMDRIVE=%DOSDRV% 

IF x%SYSTEMDRIVE%x==xx SET SYSTEMDRIVE=C: 

IF NOT x%TEMP%x==xx GOTO 2_Weiter 

IF NOT x%TMP%x==xx GOTO 1_Weiter 

MKDIR %SYSTEMDRIVE%\TEMP 

SET TEMP=%SYSTEMDRIVE%\TEMP 

GOTO 2_Weiter 

:1_Weiter 

SET TEMP=%TMP% 

:2_Weiter 

SET QBASICTEMP=%TEMP% 

IF NOT EXIST INSTALL.BAT GOTO 1_Fehler 

CALL INSTALL.BAT 

 

REM Das Programm ist auf die CODEPAGE 437 ausgerichtet: 

REM --------------------------------------------------- 

IF x%WINDIR%x==xx GOTO 1_DosPfad 

IF EXIST %WINDIR%\SYSTEM32\EGA.CPI GOTO WindowsNT 

IF EXIST %WINDIR%\COMMAND\EGA.CPI GOTO Windows95 

:1_DosPfad 

IF EXIST %SYSTEMDRIVE%\DOS\EGA.CPI GOTO Dos 

IF EXIST %DOSDIR%\CPI\EGA.CPX Goto FreeDos 

KEYB GR 437 

IF NOT EXIST Z:\COMMAND.COM GOTO Notfall 

IF NOT EXIST Z:\MOUNT.COM GOTO Notfall 

IF NOT EXIST Z:\INTRO.COM GOTO Notfall 

GOTO Start 

:Notfall 

ECHO Datei EGA.CPI wurde nicht gefunden. 

ECHO Darum wird nun die CODEPAGE 437 voraus gesetzt. 

PAUSE 

GOTO Start 

:Dos 

MODE CON CODEPAGE PREPARE=((437) %SYSTEMDRIVE%\DOS\EGA.CPI) 

KEYB GR,437,%SYSTEMDRIVE%\DOS\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=437 

GOTO Start 

:FreeDos 

SET LANG=DE 

SET DIRCMD=/Y 

LH DISPLAY CON=(EGA,,1) 

MODE CON CP PREP=((437) %DOSDIR%\CPI\EGA.CPX) 

MODE CON CP SEL=437 

MODE CON CP REFRESH 

MKEYB GR 

GOTO Start 

:Windows95 

MODE CON CODEPAGE PREPARE=((437) %WINDIR%\COMMAND\EGA.CPI) 

IF EXIST %WINDIR%\COMMAND\KEYB.COM KEYB GR,437,%WINDIR%\COMMAND\KEYBOARD.SYS 

IF NOT EXIST %WINDIR%\COMMAND\KEYB.COM KEYBOARD 

GR,437,%WINDIR%\COMMAND\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=437 

GOTO Start 

:WindowsNT 

MODE CON CODEPAGE PREPARE=((437) %WINDIR%\SYSTEM32\EGA.CPI) 

REM KEYB GR,437,%WINDIR%\SYSTEM32\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=437 

CHCP 437 

 

REM Eigentlicher Programmstart: 

REM --------------------------- 

:Start 

IF NOT EXIST HISTORIK.BAS GOTO 2_Fehler 

ECHO Versuch, das Programm ″QBASIC.EXE″ zu laden: 

QBASIC.EXE /H /RUN HISTORIK.BAS 

 

REM Eventuell Dateien ausdrucken: 

REM ----------------------------- 

IF x%HISTORIKTEMP%x==xx GOTO 3_Weiter 

IF NOT EXIST %HISTORIKTEMP%\DRUCKEN.JA GOTO Vergleich 

ECHO Bitte Drucker startklar machen. 

PAUSE 

ECHO Ausdruck klein und doppelt: Eigene Routine verwenden! 

@ECHO ON 

DRUCKE %HISTORIKTEMP%\*.DR* K D 

@ECHO OFF 

DEL %HISTORIKTEMP%\DRUCKEN.JA 

PAUSE 

GOTO Start 

 

:3_Weiter 

IF x%QBASICTEMP%x==xx GOTO 4_Weiter 

IF NOT EXIST %QBASICTEMP%\HISTORIK.TMP\DRUCKEN.JA GOTO Vergleich 

ECHO Bitte Drucker startklar machen. 

PAUSE 

ECHO Ausdruck klein und doppelt: Eigene Routine verwenden! 

@ECHO ON 

DRUCKE %QBASICTEMP%\HISTORIK.TMP\*.DR* K D 

@ECHO OFF 

DEL %QBASICTEMP%\HISTORIK.TMP\DRUCKEN.JA 

PAUSE 

GOTO Start 

 

:4_Weiter 

IF NOT EXIST HISTORIK.TMP\DRUCKEN.JA GOTO Vergleich 

ECHO Bitte Drucker startklar machen. 

PAUSE 

ECHO Ausdruck klein und doppelt: Eigene Routine verwenden! 

@ECHO ON 

DRUCKE HISTORIK.TMP\*.DR* K D 

@ECHO OFF 

DEL HISTORIK.TMP\DRUCKEN.JA 

PAUSE 

GOTO Start 

 

REM Fehlerbehandlung: 

REM ----------------- 

:1_Fehler 

ECHO Das Programm ″INSTALL.BAT″ fehlt. 

PAUSE 

GOTO Ende 

:2_Fehler 

ECHO Das Programm ″HISTORIK.BAS″ fehlt. 

PAUSE 

GOTO Ende 

 

REM Zum Schluss Ergebnisse vergleichen: 

REM ----------------------------------- 

:Vergleich 

QBASIC.EXE /H /RUN VERGLEIX.BAS 

 

 

REM CODEPAGE auf Windows-Vorgabe oder anderen Wert zurücksetzen: 

REM ------------------------------------------------------------ 

:Ende 

IF x%WINDIR%x==xx GOTO 2_DosPfad 

IF EXIST %WINDIR%\SYSTEM32\EGA.CPI GOTO EndeWindowsNT 

IF EXIST %WINDIR%\COMMAND\EGA.CPI GOTO EndeWindows95 

:2_DosPfad 

IF EXIST %SYSTEMDRIVE%\DOS\EGA.CPI GOTO EndeDos 

IF EXIST %DOSDIR%\CPI\EGA.CPX GOTO EndeFreeDos 

IF NOT EXIST Z:\COMMAND.COM GOTO Schluss 

IF NOT EXIST Z:\MOUNT.COM GOTO Schluss 

IF NOT EXIST Z:\INTRO.COM GOTO Schluss 

ECHO Zum Beenden des DOS-Fensters eingeben: EXIT 

GOTO Schluss 

:EndeDos 

MODE CON CODEPAGE PREPARE=((850) %SYSTEMDRIVE%\DOS\EGA.CPI) 

KEYB GR,850,%SYSTEMDRIVE%\DOS\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=850 

GOTO Schluss 

:EndeFreeDos 

MODE CON CP PREP=((858) %DOSDIR%\CPI\EGA.CPX) 

MODE CON CP SEL=858 

MODE CON CP REFRESH 

SET DIRCMD=/P /ONG /Y 

GOTO Schluss 

:EndeWindows95 

MODE CON CODEPAGE PREPARE=((850) %WINDIR%\COMMAND\EGA.CPI) 

IF EXIST %WINDIR\COMMAND\KEYB.COM KEYB GR,850,%WINDIR%\COMMAND\KEYBOARD.SYS 

IF NOT EXIST %WINDIR%\COMMAND\KEYB.COM KEYBOARD 

GR,850,%WINDIR%\COMMAND\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=850 

GOTO Schluss 

:EndeWindowsNT 

MODE CON CODEPAGE PREPARE=((850) %WINDIR%\SYSTEM32\EGA.CPI) 

REM KEYB GR,850,%WINDIR%\SYSTEM32\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=850 

CHCP 850 

 

REM Eine Pause am Programmende sorgt für Lesbarkeit aller Fehlermeldungen: 

REM ---------------------------------------------------------------------- 

:Schluss 

PAUSE 

 

REM =========================================================================== 

REM  Ende von HISTORIK.BAT 

REM =========================================================================== 


